

Mãos na massa

Chegou a hora de você pôr em prática o que foi visto na aula. Para isso, execute os passos listados abaixo.

- 1) No projeto **ByteBank.SistemaAgencia**, acesse o código do evento de `Click` do botão `Button9`, do formulário `Ftm_TesteLista`.
- 2) Modifique-o para o código abaixo, que manipula mais características do `DataTable` e `DataSet`:

```
Private Sub Button9_Click(sender As Object, e As EventArgs) Handles Button9.Click

    ' #####
    ' Criar uma lista de contas correntes e clientes
    ' #####

    Dim Cliente As New List(Of Cliente)
    Cliente.Add(CriarCliente("123123123123", "João da Silva", "Engenheiro"))
    Cliente.Add(CriarCliente("123123129999", "Pedro Alcantara", "Médico"))
    Cliente.Add(CriarCliente("123123888888", "Luzia Medeiros", "Mecânico"))
    Cliente.Add(CriarCliente("125656565656", "Márcia Andrade", "Engenheiro"))
    Cliente.Add(CriarCliente("186868686868", "Joana Silveira", "Empresário"))
    Cliente.Add(CriarCliente("139393939393", "Marcelo Almeida", "Engenheiro"))
    Cliente.Add(CriarCliente("186868686869", "Vitor Junior", "Analista de Sistemas"))

    Dim Contas As New List(Of ContaCorrente)
    Contas.Add(New ContaCorrente(277, 123221))
    Contas(0).titular = Cliente(0)
    Contas.Add(New ContaCorrente(277, 133323))
    Contas(1).titular = Cliente(0)
    Contas.Add(New ContaCorrente(277, 244323))
    Contas(2).titular = Cliente(2)
    Contas.Add(New ContaCorrente(277, 124543))
    Contas(3).titular = Cliente(1)
    Contas.Add(New ContaCorrente(277, 543434))
    Contas(4).titular = Cliente(1)
    Contas.Add(New ContaCorrente(277, 653443))
    Contas(5).titular = Cliente(3)
    Contas.Add(New ContaCorrente(277, 126544))
    Contas(6).titular = Cliente(4)
    Contas.Add(New ContaCorrente(277, 748383))
    Contas(7).titular = Cliente(5)

    ' #####
    ' criando a tabela de contas correntes
    ' #####

    ' Conta Corrente : Agencia (Int), Conta (Int), cfp (string), saldo (double)

    Dim Dt As New DataTable
    Dt.Columns.Add(CriarColuna("Agencia", "System.Int32"))
    Dt.Columns.Add(CriarColuna("Conta", "System.Int32"))
    Dt.Columns.Add(CriarColuna("CPF", "System.String"))
    Dt.Columns.Add(CriarColuna("Saldo", "System.Double"))
```

```

' definir chave primária

Dim vArrayChavesContas(1) As DataColumn
Dt.Columns("Agencia").AllowDBNull = False
Dt.Columns("Conta").AllowDBNull = False
vArrayChavesContas(0) = Dt.Columns("Agencia")
vArrayChavesContas(1) = Dt.Columns("Conta")
Dt.PrimaryKey = vArrayChavesContas

' Nome da tabela

Dt.TableName = "Contas Correntes"

' #####
' Criar a tabela de Clientes
' #####

Dim Dt2 As New DataTable
Dt2.Columns.Add(CriarColuna("CPF", "System.String"))
Dt2.Columns.Add(CriarColuna("Nome", "System.String"))
Dt2.Columns.Add(CriarColuna("Profissao", "System.String"))

' Criando chave primária para cliente

Dim vArrayChavesCliente(0) As DataColumn
Dt2.Columns("CPF").AllowDBNull = False
vArrayChavesCliente(0) = Dt2.Columns("CPF")
Dt2.PrimaryKey = vArrayChavesCliente

' Nome da tabela

Dt2.TableName = "Clientes"

' #####
' Criar o DataSET
' #####

Dim DS As New DataSet
DS.Tables.Add(Dt)
DS.Tables.Add(Dt2)

' #####
' Criar a chave estrangeira
'
' Clientes -----> Contas Correntes (FK)
' #####

Dim FKClientesCC As New ForeignKeyConstraint("FKClientesCC",
    DS.Tables("Clientes").Columns("CPF"),
    DS.Tables("Contas Correntes").Columns("CPF"))

FKClientesCC.DeleteRule = Rule.None

DS.Tables("Contas Correntes").Constraints.Add(FKClientesCC)

' #####
' Incluir dados da tabela de clientes do DataSET

```

```

' #####

For I As Integer = 0 To Cliente.Count - 1
    Dim ClienteAtual As Cliente = Cliente(I)
    Dim Dr As DataRow = DS.Tables("Clientes").NewRow

    Dr(0) = ClienteAtual.cpf
    Dr(1) = ClienteAtual.nome
    Dr(2) = ClienteAtual.profissao

    DS.Tables("Clientes").Rows.Add(Dr)
Next

' #####
' Incluir dados da tabela de contas corrente do DataSET
' #####

For I As Integer = 0 To Contas.Count - 1
    Dim ContaAtual As ContaCorrente = Contas(I)
    Dim Dr As DataRow = DS.Tables("Contas Correntes").NewRow

    Dr(0) = ContaAtual.agencia
    Dr(1) = ContaAtual.numero
    Dr(2) = ContaAtual.titular.cpf
    Dr(3) = ContaAtual.saldo

    DS.Tables("Contas Correntes").Rows.Add(Dr)
Next

' #####
' Criação de filtro para obter contas correntes de um correntista
' #####

Dim vCritério As String = "CPF = '123123129999'"
Dim DtFiltro As DataTable = DS.Tables("Contas Correntes").Clone
Dim DrLinhas As DataRow() = DS.Tables("Contas Correntes").Select(vCritério)
For Each row As DataRow In DrLinhas
    DtFiltro.ImportRow(row)
Next

' #####
' Ordenar a tabela de clientes por nome
' #####

Dim DtView As DataView
DtView = DS.Tables("Clientes").DefaultView
Dim vCritérioOrdenacao As String = "Nome"
DtView.Sort = vCritérioOrdenacao
Dim DtOrdenado As New DataTable
DtOrdenado = DtView.ToTable

' #####
' Filtrando e ordenando ao mesmo tempo
' #####

Dim DtView2 As New DataView(DS.Tables("Contas Correntes"), "CPF = '123123129999'",
    "CPF", DataViewRowState.CurrentRows)

```

```
' #####  
' Excluindo linhas da tabela  
' #####  
  
Dim vCritério2 As String = "CPF = '123123129999'"  
Dim DrLinhas2 As DataRow() = DS.Tables("Contas Correntes").Select(vCritério2)  
For Each row As DataRow In DrLinhas  
    DS.Tables("Contas Correntes").Rows.Remove(row)  
Next  
  
End Sub
```